G02HAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G02HAF performs bounded influence regression (M-estimates). Several standard methods are available.

2 Specification

```
      SUBROUTINE G02HAF(INDW, IPSI, ISIGMA, INDC, N, M, X, IX, Y, CPSI,
     1                  H1, H2, H3, CUCV, DCHI, THETA, SIGMA, C, IC, RS,
     2                  WGT, TOL, MAXIT, NITMON, WORK, IFAIL)
      INTEGER           INDW, IPSI, ISIGMA, INDC, N, M, IX, IC, MAXIT,
     1                  NITMON, IFAIL
      real              X(IX,M), Y(N), CPSI, H1, H2, H3, CUCV, DCHI,
     1                  THETA(M), SIGMA, C(IC,M), RS(N), WGT(N), TOL,
     2                  WORK(4*N+M*(N+M))
```

3 Description

For the linear regression model

$$y = X\theta + \epsilon$$

where y is a vector of length n of the dependent variable,

X is a n by m matrix of independent variables of column rank k,

θ is a vector of length m of unknown parameters,

and ϵ is a vector of length n of unknown errors with $\text{var}(\epsilon_i) = \sigma^2$:

G02HAF calculates the M-estimates given by the solution, $\hat{\theta}$, to the equation

$$\sum_{i=1}^{n} \psi(r_i/(\sigma w_i))w_i x_{ij} = 0 \quad j = 1, 2, ..., m \tag{1}$$

where r_i is the ith residual i.e., the ith element of $r = y - X\hat{\theta}$,

ψ is a suitable weight function,

w_i are suitable weights,

and σ may be estimated at each iteration by the median absolute deviation of the residuals:

$$\hat{\sigma} = \underset{i}{\text{med}}[|r_i|]/\beta_1$$

or as the solution to:

$$\sum_{i=1}^{n} \chi(r_i/(\hat{\sigma} w_i))w_i^2 = (n-k)\beta_2$$

for suitable weight function χ, where β_1 and β_2 are constants, chosen so that the estimator of σ is asymptotically unbiased if the errors, ϵ_i, have a Normal distribution. Alternatively σ may be held at a constant value.

The above describes the Schweppe type regression. If the w_i are assumed to equal 1 for all i then Huber type regression is obtained. A third type, due to Mallows, replaces (1) by

$$\sum_{i=1}^{n} \psi(r_i/\sigma)w_i x_{ij} = 0, \quad j = 1, 2, \ldots, m.$$

This may be obtained by use of the transformations

$$w_i^* \leftarrow \sqrt{w_i}$$
$$y_i^* \leftarrow y_i\sqrt{w_i}$$
$$x_{ij}^* \leftarrow x_{ij}\sqrt{w_i}, \quad j = 1,2,\ldots,m$$

(see Marazzi Section 3).

For Huber and Schweppe type regressions, β_1 is the 75th percentile of the standard Normal distribution. For Mallows type regression β_1 is the solution to

$$\frac{1}{n}\sum_{i=1}^{n}\Phi(\beta_1/\sqrt{w_i}) = .75$$

where Φ is the standard Normal cumulative distribution function.

β_2 is given by:

$$\begin{aligned} \beta_2 &= \int_{-\infty}^{\infty}\chi(z)\phi(z)\,dz, && \text{in Huber case;} \\ \beta_2 &= \frac{1}{n}\sum_{i=1}^{n} w_i\int_{-\infty}^{\infty}\chi(z)\phi(z)\,dz, && \text{in Mallows case;} \\ \beta_2 &= \frac{1}{n}\sum_{i=1}^{n} w_i^2\int_{-\infty}^{\infty}\chi(z/w_i)\phi(z)\,dz, && \text{in Schweppe case;} \end{aligned}$$

where ϕ is the standard Normal density, i.e.,

$$\frac{1}{\sqrt{2\pi}}\exp\left(-\frac{1}{2}x^2\right).$$

The calculation of the estimates of θ can be formulated as an iterative weighted least-squares problem with a diagonal weight matrix G given by

$$G_{ii} = \begin{cases} \dfrac{\psi(r_i/(\sigma w_i))}{(r_i/(\sigma w_i))}, & r_i \neq 0 \\ \psi'(0), & r_i = 0 \end{cases}$$

where $\psi'(t)$ is the derivative of ψ at the point t.

The value of θ at each iteration is given by the weighted least-squares regression of y on X. This is carried out by first transforming the y and X by

$$\tilde{y}_i = y_i\sqrt{G_{ii}}$$
$$\tilde{x}_{ij} = x_{ij}\sqrt{G_{ii}}, \quad j = 1,2,\ldots,m$$

and then using F04JGF. If X is of full column rank then an orthogonal-triangular (QR) decomposition is used, if not, a singular value decomposition is used.

The following functions are available for ψ and χ in G02HAF.

(a) **Unit Weights**

$\psi(t) = t, \qquad \chi(t) = \frac{t^2}{2}$

this gives least-squares regression.

(b) **Huber's Function**

$$\psi(t) = \max(-c, \min(c,t)), \qquad \chi(t) = \begin{cases} \dfrac{t^2}{2}, & |t| \leq d \\ \dfrac{d^2}{2}, & |t| > d \end{cases}$$

(c) **Hampel's Piecewise Linear Function**

$$\psi_{h_1,h_2,h_3}(t) = -\psi_{h_1,h_2,h_3}(-t) = \begin{cases} t, & 0 \leq t \leq h_1 \\ h_1, & h_1 \leq t \leq h_2 \\ h_1(h_3 - t)/(h_3 - h_2), & h_2 \leq t \leq h_3 \\ 0, & h_3 < t \end{cases}$$

$$\chi(t) = \begin{cases} \frac{t^2}{2}, & |t| \leq d \\ \frac{d^2}{2}, & |t| > d \end{cases}$$

(d) **Andrew's Sine Wave Function**

$$\psi(t) = \begin{cases} \sin t, & -\pi \leq \pi \\ 0, & |t| > \pi \end{cases} \qquad \chi(t) = \begin{cases} \frac{t^2}{2}, & |t| \leq d \\ \frac{d^2}{2}, & |t| > d \end{cases}$$

(e) **Tukey's Bi-weight**

$$\psi(t) = \begin{cases} t(1-t^2)^2, & |t| \leq 1 \\ 0, & |t| > 1 \end{cases} \qquad \chi(t) = \begin{cases} \frac{t^2}{2}, & |t| \leq d \\ \frac{d^2}{2}, & |t| > d \end{cases}$$

where c, $h_{1,}h_{2,}h_3$, and d are given constants.

Several schemes for calculating weights have been proposed, see Hampel *et al.* [1] and Marazzi [3]. As the different independent variables may be measured on different scales, one group of proposed weights aims to bound a standardized measure of influence. To obtain such weights the matrix A has to be found such that:

$$\frac{1}{n}\sum_{i=1}^{n} u(\|z_i\|_2)z_i z_i^T = I$$

and

$$z_i = Ax_i$$

where x_i is a vector of length m containing the ith row of X,

A is a m by m lower triangular matrix,

and u is a suitable function.

The weights are then calculated as

$$w_i = f(\|z_i\|_2)$$

for a suitable function f.

G02HAF finds A using the iterative procedure

$$A_k = (S_k + I)A_{k-1}$$

where $S_k = (s_{jl})$,

$$s_{jl} = \begin{cases} -\min[\max(h_{jl}/n, -BL), BL] & j > 1 \\ -\min[\max(\frac{1}{2}(h_{jj}/n - 1), -BD), BD] & j = 1 \end{cases}$$

and

$$h_{jl} = \sum_{i=1}^{n} u(\|z_i\|_2)z_{ij}z_{il}$$

and BL and BD are bounds set at 0.9.

Two weights are available in G02HAF:

(i) Krasker–Welsch weights

$$u(t) = g_1\left(\frac{c}{t}\right)$$

where $g_1(t) = t^2 + (1-t^2)(2\Phi(t)-1) - 2t\phi(t)$,

$\Phi(t)$ is the standard Normal cumulative distribution function,

$\phi(t)$ is the standard Normal probability density function,

and $f(t) = \frac{1}{t}$.

These are for use with Schweppe type regression.

(ii) Maronna's proposed weights

$$u(t) = \begin{matrix} c/t^2 & |t| > c \\ 1 & |t| \leq c \end{matrix}$$

$$f(t) = \sqrt{u(t)}.$$

These are for use with Mallows type regression.

Finally the asymptotic variance-covariance matrix, C, of the estimates θ is calculated.

For Huber type regression

$$C = f_H(X^TX)^{-1}\hat{\sigma}^2$$

where

$$f_H = \frac{1}{n-m}\frac{\sum_{i=1}^n \psi^2(r_i/\hat{\sigma})}{\left(\frac{1}{n}\sum_{i=1}^n \psi'\left(\frac{r_i}{\hat{\sigma}}\right)\right)^2}\kappa^2$$

$$\kappa^2 = 1 + \frac{m}{n}\frac{\frac{1}{n}\sum_{i=1}^n\left(\psi'(r_i/\hat{\sigma}) - \frac{1}{n}\sum_{i=1}^n \psi'(r_i/\hat{\sigma})\right)^2}{\left(\frac{1}{n}\sum_{i=1}^n \psi'\left(\frac{r_i}{\hat{\sigma}}\right)\right)^2}$$

See Huber [2] and Marazzi [4].

For Mallows and Schweppe type regressions C is of the form

$$\frac{\hat{\sigma}^2}{n}S_1^{-1}S_2S_1^{-1}$$

where $S_1 = \frac{1}{n}X^TDX$ and $S_2 = \frac{1}{n}X^TPX$.

D is a diagonal matrix such that the ith element approximates $E(\psi'(r_i/(\sigma w_i)))$ in the Schweppe case and $E(\psi'(r_i/\sigma)w_i)$ in the Mallows case.

P is a diagonal matrix such that the ith element approximates $E(\psi^2(r_i/(\sigma w_i))w_i^2)$ in the Schweppe case and $E(\psi^2(r_i/\sigma)w_i^2)$ in the Mallows case.

Two approximations are available in G02HAF:

(1) Average over the r_i

Schweppe	Mallows
$D_i = \left(\frac{1}{n}\sum_{j=1}^n \psi'\left(\frac{r_j}{\hat{\sigma}w_i}\right)\right)w_i$	$D_i = \left(\frac{1}{n}\sum_{j=1}^n \psi'\left(\frac{r_j}{\hat{\sigma}}\right)\right)w_i$
$P_i = \left(\frac{1}{n}\sum_{j=1}^n \psi^2\left(\frac{r_j}{\hat{\sigma}w_i}\right)\right)w_i^2$	$P_i = \left(\frac{1}{n}\sum_{j=1}^n \psi^2\left(\frac{r_j}{\hat{\sigma}}\right)\right)w_i^2$

(2) Replace expected value by observed

Schweppe	Mallows
$D_i = \psi'\left(\frac{r_i}{\hat{\sigma}w_i}\right)w_i$	$D_i = \psi'\left(\frac{r_i}{\hat{\sigma}}\right)w_i$
$P_i = \psi^2\left(\frac{r_i}{\hat{\sigma}w_i}\right)w_i^2$	$P_i = \psi^2\left(\frac{r_i}{\hat{\sigma}}\right)w_i^2$

See Hampel *et al.* [1] and Marazzi [4].

Note: There is no explicit provision in the routine for a constant term in the regression model. However, the addition of a dummy variable whose value is 1.0 for all observations will produce a value of $\hat{\theta}$ corresponding to the usual constant term.

G02HAF is based on routines in ROBETH, see Marazzi [3].

4 References

[1] Hampel F R, Ronchetti E M, Rousseeuw P J and Stahel W A (1986) *Robust Statistics. The Approach Based on Influence Functions* Wiley

[2] Huber P J (1981) *Robust Statistics* Wiley

[3] Marazzi A (1987) Weights for bounded influence regression in ROBETH *Cah. Rech. Doc. IUMSP, No. 3 ROB 3* Institut Universitaire de Médecine Sociale et Préventive, Lausanne

[4] Marazzi A (1987) Subroutines for robust and bounded influence regression in ROBETH *Cah. Rech. Doc. IUMSP, No. 3 ROB 2* Institut Universitaire de Médecine Sociale et Préventive, Lausanne

5 Parameters

1: INDW — INTEGER *Input*

On entry: specifies the type of regression to be performed.

If INDW $= 0$, Huber type regression.
If INDW < 0, Mallows type regression with Maronna's proposed weights.
If INDW > 0, Schweppe type regression with Krasker–Welsch weights.

2: IPSI — INTEGER *Input*

On entry: specifies which ψ function is to be used.

If IPSI $= 0$, $\psi(t) = t$, i.e., least-squares.
If IPSI $= 1$, Huber's function.
If IPSI $= 2$, Hampel's piecewise linear function.
If IPSI $= 3$, Andrew's sine wave.
If IPSI $= 4$, Tukey's bi-weight.

Constraint: $0 \leq$ IPSI ≤ 4.

3: ISIGMA — INTEGER *Input*

On entry: specifies how σ is to be estimated.

If ISIGMA < 0, σ is estimated by median absolute deviation of residuals.
If ISIGMA $= 0$, σ is held constant at its initial value.
If ISIGMA > 0, σ is estimated using the χ function.

4: INDC — INTEGER *Input*

On entry: if INDW $\neq 0$, INDC specifies the approximations used in estimating the covariance matrix of $\hat{\theta}$.

If INDC $= 1$, averaging over residuals.
If INDC $\neq 1$, replacing expected by observed.
If If INDW $= 0$, INDC is not referenced.

5: N — INTEGER *Input*

On entry: the number n, of observations.

Constraint: N > 1.

6: M — INTEGER *Input*

On entry: the number m, of independent variables.

Constraint: 1 ≤ M < N.

7: X(IX,M) — ***real*** array *Input/Output*

On entry: the values of the X matrix, i.e., the independent variables. $X(i,j)$ must contain the ijth element of X, for $i = 1, 2, \ldots, n$; $j = 1, 2, \ldots, m$.

If INDW < 0, then during calculations the elements of X will be transformed as described in Section 3. Before exit the inverse transformation will be applied. As a result there may be slight differences between the input X and the output X.

On exit: unchanged, except as described above.

8: IX — INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G02HAF is called.

Constraint: IX ≥ N.

9: Y(N) — ***real*** array *Input/Output*

On entry: the data values of the dependent variable.

Y(i) must contain the value of y for the ith observation, for $i = 1, 2, \ldots, n$.

If INDW < 0, then during calculations the elements of Y will be transformed as described in Section 3. Before exit the inverse transformation will be applied. As a result there may be slight differences between the input Y and the output Y.

On exit: unchanged, except as described above.

10: CPSI — ***real*** *Input*

On entry: if IPSI = 1, CPSI must specify the parameter, c, of Huber's ψ function. If IPSI ≠ 1 on entry, CPSI is not referenced.

Constraint: if IPSI = 1, then CPSI > 0.0.

11: H1 — ***real*** *Input*
12: H2 — ***real*** *Input*
13: H3 — ***real*** *Input*

On entry: if IPSI = 2, H1, H2, and H3 must specify the parameters h_1, h_2, and h_3, of Hampel's piecewise linear ψ function. H1, H2, and H3 are not referenced if IPSI ≠ 2.

Constraint: if IPSI = 2, then 0.0 ≤ H1 ≤ H2 ≤ H3 and H3 > 0.0.

14: CUCV — ***real*** *Input*

On entry:

If INDW < 0, CUCV must specify the value of the constant, c, of the function u for Maronna's proposed weights.
If INDW > 0, CUCV must specify the value of the function u for the Krasker–Welsch weights.
If INDW = 0, CUCV is not referenced.

Constraints:

if INDW < 0, CUCV ≥ M.
if INDW > 0, CUCV ≥ $\sqrt{\text{M}}$.

15: DCHI — ***real*** *Input*

On entry: the constant, d, of the χ function. DCHI is not referenced if IPSI = 0, or if ISIGMA ≤ 0.

Constraint: if IPSI $\neq 0$ and ISIGMA > 0, DCHI > 0.0.

16: THETA(M) — ***real*** array *Input/Output*

On entry: starting values of the parameter vector θ. These may be obtained from least-squares regression. Alternatively if ISIGMA < 0 and SIGMA $= 1$ or if ISIGMA > 0 and SIGMA approximately equals the standard deviation of the dependent variable, y, then THETA(i) $= 0.0$, for $i = 1, 2, \ldots, m$ may provide reasonable starting values.

On exit: THETA(i) contains the M-estimate of θ_i, for $i = 1, 2, \ldots, m$.

17: SIGMA — ***real*** *Input/Output*

On entry: a starting value for the estimation of σ. SIGMA should be approximately the standard deviation of the residuals from the model evaluated at the value of θ given by THETA on entry.

Constraint: SIGMA > 0.0.

On exit: SIGMA contains the final estimate of σ if ISIGMA $\neq 0$ or the value assigned on entry if ISIGMA $= 0$.

18: C(IC,M) — ***real*** array *Output*

On exit: the diagonal elements of C contain the estimated asymptotic standard errors of the estimates of θ, i.e., C(i,i) contains the estimated asymptotic standard error of the estimate contained in THETA(i).

The elements above the diagonal contain the estimated asymptotic correlation between the estimates of θ, i.e., C(i,j), $1 \leq i < j \leq m$ contains the asymptotic correlation between the estimates contained in THETA(i) and THETA(j).

The elements below the diagonal contain the estimated asymptotic covariance between the estimates of θ, i.e., C(i,j), $1 \leq j < i \leq m$ contains the estimated asymptotic covariance between the estimates contained in THETA(i) and THETA(j).

19: IC — INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which G02HAF is called.

Constraint: IC $\geq$ M.

20: RS(N) — ***real*** array *Output*

On exit: the residuals from the model evaluated at final value of THETA i.e., RS contains the vector $(y - X\hat{\theta})$.

21: WGT(N) — ***real*** array *Output*

On exit: the vector of weights. WGT(i) contains the weight for the ith observation, for $i = 1, 2, \ldots, n$.

22: TOL — ***real*** *Input*

On entry: the relative precision for the calculation of A (if INDW $\neq 0$), the estimates of θ and the estimate of σ (if ISIGMA $\neq 0$). Convergence is assumed when the relative change in all elements being considered is less than TOL.

If INDW < 0 and ISIGMA < 0, TOL is also used to determine the precision of β_1.

It is advisable for TOL to be greater than 100****machine precision***.

Constraint: TOL > 0.0.

23: MAXIT — INTEGER *Input*

On entry: the maximum number of iterations that should be used in the calculation of A (if INDW $\neq 0$), and of the estimates of θ and σ, and of β_1 (if INDW < 0 and ISIGMA < 0).

A value of MAXIT $= 50$ should be adequate for most uses.

Constraint: MAXIT > 0.

24: NITMON — INTEGER *Input*

On entry: the amount of information that is printed on each iteration.

If NITMON $= 0$ no information is printed.
If NITMON $\neq 0$ the current estimate of θ, the change in θ during the current iteration and the current value of σ are printed on the first and every ABS(NITMON) iterations.

Also, if INDW $\neq 0$ and NITMON > 0 then information on the iterations to calculate A is printed. This is the current estimate of A and the maximum value of S_{ij} (see Section 3).

When printing occurs the output is directed to the current advisory message unit (see X04ABF).

25: WORK(4∗N+M∗(N+M)) — ***real*** array *Output*

On exit: the following values are assigned to WORK:

WORK(1) $= \beta_1$ if ISIGMA < 0, or WORK(1) $= \beta_2$ if ISIGMA > 0.
WORK(2) = number of iterations used to calculate A.
WORK(3) = number of iterations used to calculate final estimates of θ and σ.
WORK(4) $= k$, the rank of the weighted least-squares equations.

The rest of the array is used as workspace.

26: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL $\neq 0$ on exit, users are recommended to set IFAIL to -1 before entry. **It is then essential to test the value of IFAIL on exit**.

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL $= 1$

On entry, N ≤ 1,
or M < 1,
or N $\leq$ M,
or IX $<$ N,
or IC $<$ M.

IFAIL $= 2$

On entry, IPSI < 0,
or IPSI > 4.

IFAIL = 3

On entry, SIGMA $\leq$ 0.0,
or IPSI = 1 and CPSI $\leq$ 0.0,
or IPSI = 2 and H1 $<$ 0.0,
or IPSI = 2 and H1 $>$ H2,
or IPSI = 2 and H2 $>$ H3,
or IPSI = 2 and H1 = H2 = H3 = 0.0,
or IPSI $\neq$ 0 and ISIGMA $>$ 0 and DCHI $\leq$ 0.0,
or INDW $>$ 0 and CUCV $< \sqrt{\mathrm{M}}$,
or INDW $<$ 0 and CUCV $<$ M.

IFAIL = 4

On entry, TOL $\leq$ 0.0,
or MAXIT $\leq$ 0.

IFAIL = 5

The number of iterations required to calculate the weights exceeds MAXIT. (Only if INDW $\neq$ 0).

IFAIL = 6

The number of iterations required to calculate β_1 exceeds MAXIT. (Only if INDW $<$ 0 and ISIGMA $<$ 0).

IFAIL = 7

Either the number of iterations required to calculate θ and σ exceeds MAXIT (NOTE, in this case WK(3) = MAXIT on exit), or the iterations to solve the weighted least-squares equations failed to converge. The latter is an unlikely error exit.

IFAIL = 8

The weighted least-squares equations are not of full rank.

IFAIL = 9

If INDW = 0 then $(X^T X)$ is almost singular.

If INDW $\neq$ 0 then S_1 is singular or almost singular. This may be due to too many diagonal elements of the matrix being zero, see Section 8.

IFAIL = 10

In calculating the correlation factor for the asymptotic variance-covariance matrix either the value of

$$\frac{1}{n}\sum_{i=1}^{n} \psi'(r_i/\hat{\sigma}) = 0,$$

or

$$\kappa = 0,$$

or

$$\sum_{i=1}^{n} \psi^2(r_i/\hat{\sigma}) = 0.$$

See Section 8. In this case C is returned as $X^T X$.

(Only if INDW = 0).

IFAIL = 11

The estimated variance for an element of $\theta \leq 0$.

In this case the diagonal element of C will contain the negative variance and the above diagonal elements in the row and column corresponding to the element will be returned as zero.

This error may be caused by rounding errors or too many of the diagonal elements of P being zero. See Section 8.

IFAIL = 12

The degrees of freedom for error, $n - k \leq 0$ (this is an unlikely error exit), or the estimated value of σ was 0 during an iteration.

7 Accuracy

The precision of the estimates is determined by TOL, see Section 5. As a more stable method is used to calculate the estimates of θ than is used to calculate the covariance matrix, it is possible for the least-squares equations to be of full rank but the $(X^T X)$ matrix to be too nearly singular to be inverted.

8 Further Comments

In cases when ISIGMA ≥ 0 it is important for the value of SIGMA to be of a reasonable magnitude. Too small a value may cause too many of the winsorised residuals, i.e., $\psi(r_i/\sigma)$ to be zero or a value of $\psi'(r_i/\sigma)$, used to estimate the asymptotic covariance matrix, to be zero. This can lead to errors IFAIL = 8, 9 (if INDW $\neq 0$), 10 (if INDW $= 0$) and 11.

G02HBF, G02HDF and G02HFF together carry out the same calculations as G02HAF but for user-supplied functions for ψ, χ, ψ' and u.

9 Example

The number of observations and the number of x variables are read in followed by the data. The option parameters are then read in (in this case giving: Schweppe type regression with Hampel's ψ function and Huber's χ function and then using the 'replace expected by observed' option in calculating the covariances). Finally a set of values for the constants are read in.

After a call to G02HAF, $\hat{\theta}$, its standard error and $\hat{\sigma}$ are printed. In addition the weight and residual for each observation is printed.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G02HAF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, MMAX
      PARAMETER        (NMAX=8,MMAX=3)
*     .. Local Scalars ..
      real             CPSI, CUCV, DCHI, H1, H2, H3, SIGMA, TOL
      INTEGER          I, IC, IFAIL, INDC, INDW, IPSI, ISIGMA, IX, J, M,
     +                 MAXIT, N, NITMON
*     .. Local Arrays ..
      real             C(MMAX,MMAX), RS(NMAX), THETA(MMAX), WGT(NMAX),
```

```
     +                 WORK(4*NMAX+MMAX*(NMAX+MMAX)), X(NMAX,MMAX),
     +                 Y(NMAX)
*     .. External Subroutines ..
      EXTERNAL         G02HAF, X04ABF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G02HAF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      CALL X04ABF(1,NOUT)
*     Read in number of observations and number of X variables
      READ (NIN,*) N, M
      WRITE (NOUT,*)
      IF (N.GT.0 .AND. N.LE.NMAX .AND. M.GT.0 .AND. M.LE.MMAX) THEN
*        Read in X and Y
         DO 20 I = 1, N
            READ (NIN,*) (X(I,J),J=1,M), Y(I)
   20    CONTINUE
*        Read in control parameters
         READ (NIN,*) INDW, IPSI, ISIGMA
*        Read in appropriate weight function parameters.
         IF (INDW.NE.0) READ (NIN,*) CUCV, INDC
         IF (IPSI.GT.0) THEN
            IF (IPSI.EQ.1) READ (NIN,*) CPSI
            IF (IPSI.EQ.2) READ (NIN,*) H1, H2, H3
            IF (ISIGMA.GT.0) READ (NIN,*) DCHI
         END IF
*        Set values of remaining parameters
         IX = NMAX
         IC = MMAX
         TOL = 0.5e-4
         MAXIT = 50
*        * Change NITMON to a positive value if monitoring information
*          is required *
         NITMON = 0
         SIGMA = 1.0e0
         DO 40 I = 1, M
            THETA(I) = 0.0e0
   40    CONTINUE
         IFAIL = -1
*
         CALL G02HAF(INDW,IPSI,ISIGMA,INDC,N,M,X,IX,Y,CPSI,H1,H2,H3,
     +               CUCV,DCHI,THETA,SIGMA,C,IC,RS,WGT,TOL,MAXIT,NITMON,
     +               WORK,IFAIL)
*
         IF ((IFAIL.NE.0) .AND. (IFAIL.LT.7)) THEN
            WRITE (NOUT,99999) 'G02HAF fails, IFAIL = ', IFAIL
         ELSE
            IF (IFAIL.GE.7) THEN
               WRITE (NOUT,99999) 'G02HAF returned IFAIL = ', IFAIL
               WRITE (NOUT,*)
     +           '       Some of the following results may be unreliable'
            END IF
            WRITE (NOUT,99998) 'Sigma = ', SIGMA
            WRITE (NOUT,*)
            WRITE (NOUT,*) '       THETA      Standard'
            WRITE (NOUT,*) '                   errors'
            DO 60 J = 1, M
               WRITE (NOUT,99997) THETA(J), C(J,J)
```

```
   60       CONTINUE
            WRITE (NOUT,*)
            WRITE (NOUT,*) '     Weights     Residuals'
            DO 80 I = 1, N
               WRITE (NOUT,99997) WGT(I), RS(I)
   80       CONTINUE
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,I2)
99998 FORMAT (1X,A,F10.4)
99997 FORMAT (1X,F12.4,F13.4)
      END
```

9.2 Program Data

```
G02HAF Example Program Data

 8  3            : N   M

 1. -1. -1. 2.1  : X1 X2 X3 Y
 1. -1.  1. 3.6
 1.  1. -1. 4.5
 1.  1.  1. 6.1
 1. -2.  0. 1.3
 1.  0. -2. 1.9
 1.  2.  0. 6.7
 1.  0.  2. 5.5  : End of X1 X2 X3 and Y values

 1  2  1         : INDW  IPSI  ISIGMA
 3.0  0          : CUCV  INDC (Only if INDW.ne.0)
 1.5 3.0 4.5     : H1  H2  H3 (Only if IPSI.eq.2)
 1.5             : DCHI (Only if IPSI.gt.0 .and. ISIGMA.gt.1)
```

9.3 Program Results

```
 G02HAF Example Program Results

 Sigma =     0.2026

        THETA      Standard
                    errors
       4.0423       0.0384
       1.3083       0.0272
       0.7519       0.0311

      Weights     Residuals
       0.5783       0.1179
       0.5783       0.1141
       0.5783      -0.0987
       0.5783      -0.0026
       0.4603      -0.1256
       0.4603      -0.6385
       0.4603       0.0410
       0.4603      -0.0462
```